To appear in POPL 2007

Lazy Multivariate Higher-Order Forward-Mode AD

Barak A. Pearlmutter

Hamilton Institute NUI Maynooth, Ireland barak@cs.nuim.ie

Jeffrey Mark Siskind

School of Electrical and Computer Engineering Purdue University, USA qobi@purdue.edu

Abstract

A method is presented for computing all higher-order partial derivatives of a multivariate function R^n → R. This method works by evaluating the function under a nonstandard interpretation, lifting reals to multivariate power series. Multivariate power series, with potentially an infinite number of terms with nonzero coefficients, are represented using a lazy data structure constructed out of linear terms. A complete implementation of this method in SCHEME is presented, along with a straightforward exposition, based on Taylor expansions, of the method’s correctness.

Categories and Subject Descriptors G.1.4 [Quadrature and Numerical Differentiation]: Automatic differentiation; D.3.2 [Language Classifications]: Applicative (functional) languages

General Terms Algorithms, Languages

Keywords Power series, Nonstandard interpretation

1. Introduction

Forward-Mode Automatic Differentiation, or forward AD, [1] is a method for adapting a program that computes a function to yield one that computes its derivatives. Karczmarczuk [2–5] presented an implementation of forward AD in HASKELL. This implementation had a novel characteristic: it adapted a program that computed a univariate function f : R → R to yield one that produced an infinite stream of higher-order derivatives (f(x), f^′(x), f^{′′}(x), . . .). However, Karczmarczuk provided the details for his method only for univariate functions. Karczmarczuk [4] hinted at a generalization to multivariate functions but did not provide the details. Here, we present the details of a novel generalization of Karczmarczuk’s method to the multivariate case. In part, we use methods previously developed for implementing nestable first-order forward AD in a functional framework [6]. The crucial additional insight here, both for developing the extension and for demonstrating its correctness, involves reformulating Karczmarczuk’s method using Taylor expansions instead of the chain rule. This requires dealing with the requisite factorial factors.

Permission to make digital or hard copies of all or part of this work for personal or classroom use is granted without fee provided that copies are not made or distributed for profit or commercial advantage and that copies bear this notice and the full citation on the first page. To copy otherwise, to republish, to post on servers or to redistribute to lists, requires prior specific permission and/or a fee. POPL’07 January 17–19, 2007, Nice, France. Copyright © 2007 ACM 1-59593-575-4/07/0001. . . $5.00

2. Univariate First-Order Forward AD

The Taylor expansion [7] of f(c + ε) with respect to ε is

f(c + ε) = ∞X i=0 1 i! d^if(x) dx^i ˛ ˛ ˛_{x=c} ε^i

This implies that one can compute the i-th derivative of a univariate function f at a scalar point c by evaluating by evaluating f(c + ε) under a nonstandard interpretation replacing real numbers with univariate power series in ε, extracting the coefficient of ε^i in the result, and multiplying this by i!. Traditional forward AD [1] truncates the Taylor expansions at i > 1, thus computing a representation that contains only the first derivative.

Such truncated Taylor expansions are dual numbers [8]. We denote a dual number p as x + x^′ε, by analogy with the standard notation a + bi for complex numbers. Just as arithmetic on complex numbers a + bi can be defined by taking i^2 = −1, arithmetic on dual numbers x + x^′ε can be defined by taking ε^2 = 0 but ε ≠ 0. Furthermore, just as implementations of complex arithmetic typically represent complex numbers a + bi as Argand pairs ⟨a, b⟩, implementations of forward AD typically represent dual numbers x+x^′ε as tangent-bundle pairs ⟨x, x^′⟩. Finally, just as implementations of complex arithmetic typically overload the arithmetic primitives to manipulate complex numbers, implementations of forward AD typically overload the arithmetic primitives to manipulate dual numbers.

We use E ε p to denote the coefficient of ε in the dual number p

E ε (x + x^′ε)^△= x^′ (1)

and D f c to denote the value of the first derivative of a univariate function f at a scalar point c. Forward AD computes D f c by evaluating f (c + ε) under a nonstandard interpretation replacing real numbers with dual numbers and extracting the coefficient of ε in the result.

D f c^△= E ε (f (c + ε)) (2) The εs introduced by nested invocations of D must be distinct [6].

To see how this works, let us manually apply the mechanism to a simple example: computing the first derivative of f(x) = x^4 +2x^3 at x = 3. To do this, we first evaluate f(3 + ε).

f(3 + ε) = (3 + ε)^4 + 2(3 + ε)^3 = (81 + 108ε) + 2(27 + 27ε)

= 135 + 162ε

From this we can extract the derivative: E ε (135 + 162ε) = 162.

df(x) dx ˛ ˛ ˛ x=3 = 4x^3 + 6x^2 ˛ ˛_{x=3} = 162 = E ε (f (3 + ε))

Note that the above makes use of the restriction that ε^2 = 0 when evaluating (3+ε)^3 = 27+27ε and (3+ε)^4 = 81+108ε, dropping the ε^2, ε^3, and ε^4 terms. This is the essence of traditional forward AD when limited to the case of univariate derivatives.

3. Univariate Higher-Order Forward AD

While the above nominally computes only first derivatives, straightforward repeated application of D allows computation of higherorder derivatives. We use D f to denote partial application of D, i.e., λc . (D f c), and D^i f c to denote the value of the i-th derivative of a univariate function f at a scalar point c.

D^0 f^△= f (3)

D^i f^△= D^{i−1} (D f) when i > 0 (4)

We refer to the above method for computing higher-order derivatives as the repetition method.

Karczmarczuk [2–5] presented an alternate method for computing higher-order univariate derivatives. His method can be viewed as computing non-truncated Taylor expansions,^1 removing the restriction that ε^2 = 0, and generalizing dual numbers to univariate power series in ε. To accomplish this, we first extend the definition of E from (1) so that E ε^i p yields i! times the coefficient of ε^i in the power series p.

E ε^0 p^△= R ε p (5)

E ε^i p^△= i × E ε^{i−1} (Q ε p) when i > 0 (6)

In the above and throughout this paper, Q ε p and R ε p destructure a power series: p = (R ε p) + (Q ε p)ε. Given the above, Karczmarczuk’s method can be viewed as computing D^i f c by evaluating f (c + ε) under a nonstandard interpretation replacing real numbers with univariate power series in ε, extracting the coefficient of ε^i in the result, and multiplying this by i!.

D^i f c^△= E ε^i (f (c + ε)) (7)

To see how this works, let us manually apply the mechanism to a simple example: computing all of the higher-order derivatives of f(x) = x^4 + 2x^3 at x = 3. To do this, we first evaluate f(3 + ε).

f(3 + ε) = (3 + ε)^4 + 2(3 + ε)^3

= (81 + 108ε + 54ε^2 + 12ε^3 + ε^4) + 2(27 + 27ε + 9ε^2 + ε^3) = 135 + 162ε + 72ε^2 + 14ε^3 + ε^4

From this we can extract all of the higher-order derivatives.

f(x) ˛ ˛ ˛ ˛_{x=3}= x^4 + 2x^3 ˛ ˛ ˛ ˛_{x=3}= 135 = 0! × 135 = E ε^0 (f (3 + ε))

df(x) dx ˛ ˛ ˛ ˛_{x=3}= 4x^3 + 6x^2 ˛ ˛ ˛ ˛_{x=3}= 162 = 1! × 162 = E ε^1 (f (3 + ε)) . . d^4f(x)

dx^4 ˛ ˛ ˛ ˛_{x=3}= 24 ˛ ˛ ˛ ˛_{x=3}= 24 = 4! × 1 = E ε^4 (f (3 + ε))

4. Lazy Univariate Higher-Order Forward AD

The input to the nonstandard interpretation will always be a polynomial c + ε, an entity with a finite number of terms with nonzero coefficients. In the above example, the output of the nonstandard interpretation was also a polynomial. However some functions, such

1 As discussed in section 7, this is a reformulation of Karczmarczuk’s method which was originally formulated using the chain rule.

as sin, yield a result entity with an infinite number of terms with nonzero coefficients, i.e., a power series, even when applied to an input polynomial. Karczmarczuk addressed this by representing univariate power series as lazy unidimensional streams. We refer to this as a tower method. Karczmarczuk presented the details of his tower method only for univariate functions. The remainder of this paper presents the details of a novel tower method that generalizes to the multivariate case.

5. Multivariate Higher-Order Forward AD

We use D^{(i1,...,in)} f (c1, . . . , cn) to denote the value of

∂^{i1+···+in} f(x1, . . . , xn)

∂x1^{i1} · · · ∂xn^{in} ˛ ˛ ˛_{x1=c1,...,xn=cn}

i.e., the value of a higher-order partial derivative of a multivariate function f at a multidimensional point (c1, . . . , cn). One can generalize the repetition method to the multivariate case.

D^{(0,...,0)} f^△= f (8)

D^{(i1,...,in)} f^△= (9)

D^{(i1,...,iℓ−1,iℓ−1,iℓ+1,...,in)} λ(c1, . . . , cn) . D (λu . f (c1, . . . , cℓ−1, u, cℓ+1, . . . , cn)) cℓ when iℓ > 0

Again, each nested invocation of D must use a distinct ε [6].

One can formulate a multivariate tower method by generalizing univariate power series to multivariate power series. To accomplish this, we first note that the multivariate Taylor expansion of f((c1 + ε1), . . . , (cn + εn)) with respect to (ε1, . . . , εn) is ∞X i1=0

· · · ∞X in=0 1 i1! · · · in!

∂^{i1+···+in} f(x1, . . . , xn) ∂x1^{i1} · · · ∂xn^{in} ˛ ˛ ˛ x1=c1,...,xn=cn

ε^{i1}_1 · · · ε^{in}_n

We therefore extend the definition of E from (5–6) to the multivariate case so that E ε1^{i1} · · · εn^{in} p yields i1! · · · in! times the coefficient of ε^{i1}_1 · · · ε^{in}_n in the power series p.

E 1 p^△= p (10)

E ε1^{i1} · · · εn^{in} p^△= E ε2^{i2} · · · εn^{in} (E ε1^{i1} p) (11)

Given the above, one can compute D^{(i1,...,in)} f (c1, . . . , cn) by evaluating f ((c1 + ε1), . . . , (cn + εn)) under a nonstandard interpretation replacing real numbers with multivariate power series in distinct ε1, . . . , εn, extracting the coefficient of ε1^{i1} · · · εn^{in} in the result, and multiplying this by i1! · · · in!.

D^{(i1,...,in)} f (c1, . . . , cn)^△= (12)

E ε1^{i1} · · · εn^{in} (f ((c1 + ε1), . . . , (cn + εn)))

In the above, the εℓ must be distinct from each other and from any other ε used by nested invocations of any form of D [6].

To see how this works, let us manually apply the mechanism to a simple example: computing all of the higher-order partial derivatives of g(x, y) = x^3y + x^2y^2 at x = 2 and y = 3. To do this, we first evaluate g((2 + εx), (3 + εy )).

g((2 + εx), (3 + εy )) = (2 + εx)^3(3 + εy ) + (2 + εx)^2(3 + εy )^2 = (8 + 12εx + 6ε^2_x + ε^3_x)(3 + εy )

+ (4 + 4εx + ε^2_x)(9 + 6εy + ε^2_y ) = (24 + 36εx + 8εy + 18ε^2_x + 12εxεy + 3ε^3_x + 6ε^2_xεy + ε^3_xεy ) + (36 + 36εx + 24εy + 9ε^2_x + 24εxεy + 4ε^2_y + 6ε^2_xεy + 4εxε^2_y + ε^2_xε^2_y ) = (60 + 72εx + 32εy + 27ε^2_x + 36εxεy + 4ε^2_y + 3ε^3_x + 12ε^2_xεy + 4εxε^2_y + ε^3_xεy + ε^2_xε^2_y )

From this we can extract all of the higher-order partial derivatives.

g(x, y) ˛ ˛ ˛ ˛_{x=2,y=3}

= x^3y + x^2y^2 ˛ ˛ ˛ ˛_{x=2,y=3}

= 60 = 0! × 0! × 60

= E ε^0_xε^0_y (g ((2 + εx), (3 + εy )))

∂g(x, y) ∂x ˛ ˛ ˛_{x=2,y=3}

= 3x^2y + 2xy^2 ˛ ˛ ˛_{x=2,y=3}

= 72 = 1! × 0! × 72

= E ε^1_xε^0_y (g ((2 + εx), (3 + εy ))) . . ∂^4g(x, y) ∂x^3∂y ˛ ˛ ˛ ˛_{x=2,y=3} = 6 ˛ ˛ ˛ ˛_{x=2,y=3}

= 6 = 3! × 1! × 1

= E ε^3_xε^1_y (g ((2 + εx), (3 + εy )))

∂^4g(x, y) ∂x^2∂y^2 ˛ ˛ ˛_{x=2,y=3} = 4 ˛ ˛ ˛_{x=2,y=3}

= 4 = 2! × 2! × 1

= E ε^2_xε^2_y (g ((2 + εx), (3 + εy )))

Two difficulties arise when attempting to implement the above. First, there is the need to maintain the distinction between the different εs, addressed in previous work [6]. Second, there is a need to generalize lazy unidimensional streams to the multidimensional case to represent multivariate power series. We address this in the next section.

6. Lazy Multivariate Higher-Order Forward AD

A univariate polynomial x0 +x1ε+x2ε^2 +· · ·+xn−1ε^{n−1} +xnε^n can be evaluated using Horner’s method [9] as

x0 + (x1 + (x2 + · · · + (xn−1 + xnε)ε · · · )ε)ε

This indicates that univariate polynomials can be represented as nested dual numbers. Multivariate polynomials can be represented as nested tagged dual numbers, i.e., triples of the form ⟨ε, x, x^′⟩, with potentially distinct εs, to represent x + x^′ε. We assume that there is a total order ≺ over the εs and refer to such tagged dual numbers as linear terms. Power series can be represented as binary trees whose nodes are linear terms with a lazy x^′ slot. As illustrated in the code accompanying this paper, we constrain such representations to maintain the following invariants:

I-1 In any linear term x + x^′ε, the x slot is either real, or a linear term over ε^′ where ε^′ ≺ ε.

I-2 In any linear term x + x^′ε, the x^′ slot is either real, a linear term over ε^′ where ε^′ ≺ ε, or a linear term over ε.

These ensure that the coefficient of each term in a multivariate series is stored in at most one leaf. They also simplify proofs of termination of the derivative-taking code and lower the time bound on access to a multivariate power series.

Figure 1 contains a SCHEME implementation of an API for manipulating multivariate power series represented as lazy linear terms. Central to this API are mechanisms Q ε p and R ε p for computing the quotient and remainder when dividing the power series p by the variable ε.

R ε r^△= r when r ∈ R (13)

R ε (x + x^′ε^′)^△= x + x^′ε^′ when ε^′ ≺ ε (14)

R ε (x + x^′ε)^△= x (15)

R ε (x + x^′ε^′)^△= (R ε x) + (R ε x^′)ε^′ when ε ≠ ε^′ (16)

Q ε r^△= 0 when r ∈ R (17)

Q ε (x + x^′ε^′)^△= 0 when ε^′ ≺ ε (18)

Q ε (x + x^′ε)^△= x^′ (19)

Q ε (x + x^′ε^′)^△= (Q ε x) + (Q ε x^′)ε^′ when ε ≠ ε^′ (20)

Cases (14) and (18) and the simplifications in cases (15) and (19) are valid because of the invariants. Note that, unlike an analogous earlier API [6], the above correctly handles linear terms x + x^′ε where x^′ may itself be a linear term in the same ε. Also note that, because of laziness, unlike that earlier API, there is no easy way to implement the simplification rule x + 0ε ⇝ x.

To perform the nonstandard interpretation, we need to extend the numeric primitives to apply to power series. For simplicity, we do this only for univariate and bivariate primitives. We can assume that the power series arguments to such primitives take the form of a linear term x + x^′ε. The extension of a univariate primitive f applied to a linear term x + x^′ε can be derived via a univariate Taylor expansion about x in terms of x^′ε.

f(x + x^′ε) = ∞X i=0 f^{(i)}(x) i! (x^′ε)^i

= f(x) + ∞X i=1 f^{(i)}(x)x^{′i} i! ε^i

= f(x) +

„_∞X i=1 f^{(i)}(x)x^{′i−1} i! ε^{i−1} « x^′ε

= f(x) +

„_∞X i=0 f^{(i+1)}(x)x^{′i} (i + 1)!

ε^i « x^′ε (21)

In the above, f^{(i)} denotes the i-th derivative of f. Note that

f^′(x + x^′ε) = ∞X i=0 f^{(i+1)}(x)x^{′i} i! ε^i (22)

Also note that the right hand side of (22) is similar to the coefficient of x^′ε in (21), differing only in that the derivatives in the power series are divided by (i + 1)! instead of i!. The coefficient of x^′ε in (21) can be derived from the right hand side of (22) by postulating an operator C_ε0 to adjust the coefficients. ∞X i=0 f^{(i+1)}(x)x^{′i} (i + 1)!

ε^i = C_ε0 ∞X i=0 f^{(i+1)}(x)x^{′i} i! ε^i (23)

(As a formal power series operator, C_ε0 f(ε) =^1_ε

R_ε 0 f(ε) dε.) This operator can be defined as follows:

C_εi r^△= r i + 1 when r ∈ R (24)

C_εi (x + x^′ε)^△= (C_εi x) + (C_εi+1 x^′)ε (25)

C_εi (x + x^′ε^′)^△= (C_εi x) + (C_εi x^′)ε^′ when ε ≠ ε^′ (26)

(define <_e <)

(define =_e =)

(define linear-term? (let ((pair? pair?)) (lambda (p) (and (pair? p) (eq? (car p) ’linear-term)))))

(define-syntax linear-term (syntax-rules () ((linear-term e x x-prime) (list ’linear-term e x (delay x-prime)))))

(define epsilon cadr)

(define (r e p)
 (cond ;; Equation (13)
       ((not (linear-term? p)) p)
       ;; Equation (14)
       ((<_e (epsilon p) e) p)
       ;; Equation (15)
       ((=_e (epsilon p) e) (caddr p))
       ;; Equation (16)
       (else (linear-term (epsilon p) (r e (caddr p)) (r e (force (cadddr p)))))))

(define (q e p)
 (cond ;; Equation (17)
       ((not (linear-term? p)) 0)
       ;; Equation (18)
       ((<_e (epsilon p) e) 0)
       ;; Equation (19)
       ((=_e (epsilon p) e) (force (cadddr p)))
       ;; Equation (20)
       (else (linear-term (epsilon p) (q e (caddr p)) (q e (force (cadddr p)))))))

(define generate-epsilon (let ((e 0)) (lambda () (set! e (+ e 1)) e)))

(define (univariate-e e i p)
 (cond ;; Equation (5)
       ((zero? i) (r e p))
       ;; Equation (6)
       (else (* i (univariate-e e (- i 1) (q e p))))))

(define (multivariate-e e i p)
 (cond ;; Equation (10)
       ((null? i) p)
       ;; Equation (11)
       (else (multivariate-e (cdr e) (cdr i) (univariate-e (car e) (car i) p)))))

Figure 1. A SCHEME implementation of an API for manipulating multivariate power series represented as lazy linear terms. Note that to support nested invocation of D, the × in (6) must be the lifted variant that works on power-series arguments. Note that generated εs never escape any equations in which they are generated, i.e., (2, 7, 12, 31–34). Thus one can improve upon the above implementation by allocating and reclaiming εs in a LIFO fashion.

Note that x^′ in (23) can contain ε. This is a problem because C_εi operates by counting instances of ε, and has no way to distinguish the εs in x^′ that should not be counted. We solve this by renaming ε to a fresh ε^′ prior to calling C_εi and renaming ε^′ back to ε in the result. For the cases that arise in this paper,^2 such renaming can be accomplished with the following:

p[ε → ε]^△= p (27)

r[ε1 → ε2]^△= r when r ∈ R (28)

(x + x^′ε^′)[ε1 → ε2]^△= x + x^′ε^′ when ε^′ ≺ ε1 (29)

(x + x^′ε1)[ε1 → ε2]^△= (30) (R ε2 x) + ((Q ε2 x) + x^′[ε1 → ε2])ε2

This yields the following method for extending univariate primitives:

f (x + x^′ε)^△= (31) (f x) + ((C_ε0 (f^′ (x + x^′ε)[ε → ε^′]))[ε^′ → ε] × x^′)ε

This requires supplying f^′, the first derivative of each univariate primitive f.

Bivariate primitives can be extended when the first argument is a linear term over ε and the second argument is either a real or a linear term over ε^′ where ε^′ ≺ ε by performing a univariate Taylor

2 Renaming is only applied in equations (31–34) and therein only in cases where ε ≺ ε^′ for any existing ε. Furthermore, (29) is valid only because of the invariants.

expansion around the first argument.

f ((x + x^′ε), y)^△= (32) (f (x, y)) + ((C_ε0 (f1 ((x + x^′ε)[ε → ε^′], y)))[ε^′ → ε] × x^′)ε

Analogously, bivariate primitives can be extended when the second argument is a linear term over ε and the first argument is either a real or a linear term over ε^′ where ε^′ ≺ ε by performing a univariate Taylor expansion around the second argument.

f (x, (y + y^′ε))^△= (33) (f (x, y)) + ((C_ε0 (f2 (x, (y + y^′ε)[ε → ε^′])))[ε^′ → ε] × y^′)ε

This requires supplying f1 and f2, the partial first derivatives of each bivariate primitive f with respect to its first and second arguments respectively.

To handle the case when both arguments are linear terms over the same ε we rename the ε in one argument to a fresh ε^′, reducing this case to either (32) or (33), and then rename ε^′ back to ε in the result.

f ((x + x^′ε), (y + y^′ε))^△= (34) (f ((x + x^′ε), (y + y^′ε)[ε → ε^′]))[ε^′ → ε]

These techniques are implemented in figure 2 and used to overload some SCHEME primitives in figure 3. Figure 4 completes the implementation. Note that the computational efficiency of this implementation relies on the fact that standard SCHEME memoizes the results of forcing promises. The code from figures 1–4 is available from http://www.bcl.hamilton.ie/~qobi/tower/.

(define (c e i p)
 (cond ;; Equation (24)
       ((not (linear-term? p)) (/ p (+ i 1)))
       ;; Equation (25)
       ((=_e (epsilon p) e) (linear-term (epsilon p) (c e i (r (epsilon p) p)) (c e (+ i 1) (q (epsilon p) p))))
       ;; Equation (26)
       (else (linear-term (epsilon p) (c e i (r (epsilon p) p)) (c e i (q (epsilon p) p))))))

(define (rename e1 e2 p)
 (cond ;; Equation (27)
       ((=_e e1 e2) p)
       ;; Equation (28)
       ((not (linear-term? p)) p)
       ;; Equation (29)
       ((<_e (epsilon p) e1) p)
       ;; Equation (30)
       ((=_e (epsilon p) e1) (linear-term e2 (r e2 (r e1 p)) (+ (q e2 (r e1 p)) (rename e1 e2 (q e1 p)))))
       (else (error "This case should never occur in this program."))))

(define (lift-real->real f df/dx)
 (letrec ((self (lambda (p)
                 (cond ;; Equation (31)
                       ((linear-term? p)
                        (let ((e (epsilon p)))
                         (linear-term e
                                      (self (r e p))
                                      (* (let ((e-prime (generate-epsilon)))
                                          (rename e-prime e (c e-prime 0 (df/dx (linear-term e-prime (r e p) (q e p))))))
                                         (q e p)))))
                       (else (f p))))))
  self))

(define (lift-real*real->real f df/dx1 df/dx2)
 (letrec ((self
           (lambda (p1 p2)
            (cond
             ;; Equation (32)
             ((and (linear-term? p1) (or (not (linear-term? p2)) (<_e (epsilon p2) (epsilon p1))))
              (let ((e1 (epsilon p1)))
               (linear-term e1
                            (self (r e1 p1) p2)
                            (* (let ((e-prime (generate-epsilon)))
                                (rename e-prime e1 (c e-prime 0 (df/dx1 (linear-term e-prime (r e1 p1) (q e1 p1)) p2))))
                               (q e1 p1)))))
             ;; Equation (33)
             ((and (linear-term? p2) (or (not (linear-term? p1)) (<_e (epsilon p1) (epsilon p2))))
              (let ((e2 (epsilon p2)))
               (linear-term e2
                            (self p1 (r e2 p2))
                            (* (let ((e-prime (generate-epsilon)))
                                (rename e-prime e2 (c e-prime 0 (df/dx2 p1 (linear-term e-prime (r e2 p2) (q e2 p2))))))
                               (q e2 p2)))))
             ;; Equation (34)
             ((and (linear-term? p1) (linear-term? p2) (=_e (epsilon p1) (epsilon p2)))
              (let ((e (epsilon p1)) (e-prime (generate-epsilon))) (rename e-prime e (self p1 (rename e e-prime p2)))))
             (else (f p1 p2))))))
  self))

(define (r* p) (if (linear-term? p) (r* (r (epsilon p) p)) p))

(define (lift-real\symbol{94}n->boolean f) (lambda ps (apply f (map r* ps))))

Figure 2. A mechanism for extending SCHEME procedures of type R → R, R × R → R, and R^n → boolean to support multivariate power series. Note that the + in (30) and the × in (31–33) must be the lifted variant that works on power-series arguments. Furthermore, f^′ in (31), f1 in (32), and f2 in (33) must internally use the lifted variants of operations that work on power-series arguments.

(define pair? (let ((pair? pair?)) (lambda (x) (and (pair? x) (not (linear-term? x))))))
(define + (lift-real*real->real + (lambda (x1 x2) 1) (lambda (x1 x2) 1)))
(define - (lift-real*real->real - (lambda (x1 x2) 1) (lambda (x1 x2) -1)))
(define * (lift-real*real->real * (lambda (x1 x2) x2) (lambda (x1 x2) x1)))
(define / (lift-real*real->real / (lambda (x1 x2) (/ 1 x2)) (lambda (x1 x2) (- 0 (/ x1 (* x2 x2))))))
(define sqrt (lift-real->real sqrt (lambda (x) (/ 1 (* 2 (sqrt x))))))
(define exp (lift-real->real exp (lambda (x) (exp x))))
(define log (lift-real->real log (lambda (x) (/ 1 x))))
(define sin (lift-real->real sin (lambda (x) (cos x))))
(define cos (lift-real->real cos (lambda (x) (- 0 (sin x)))))
(define atan (lift-real*real->real
              atan (lambda (x1 x2) (/ (- 0 x2) (+ (* x1 x1) (* x2 x2)))) (lambda (x1 x2) (/ x1 (+ (* x1 x1) (* x2 x2))))))
(define = (lift-real^n->boolean =))
(define < (lift-real^n->boolean <))
(define > (lift-real^n->boolean >))
(define <= (lift-real^n->boolean <=))
(define >= (lift-real^n->boolean >=))
(define zero? (lift-real^n->boolean zero?))
(define positive? (lift-real^n->boolean positive?))
(define negative? (lift-real^n->boolean negative?))
(define real? (lift-real^n->boolean real?))

Figure 3. Overloading some SCHEME procedures that operate on reals with extensions that support multivariate power series. Note that the overloaded +, -, *, /, and atan procedures are restricted to accept precisely two arguments.

;;; Equation (2)
(define (derivative f) (lambda (c) (let ((e (generate-epsilon))) (univariate-e e 1 (f (linear-term e c 1))))))

(define (ith-derivative-by-repetition i f)
 (cond ;; Equation (3)
       ((zero? i) f)
       ;; Equation (4)
       (else (ith-derivative-by-repetition (- i 1) (derivative f)))))

;;; Equation (7)
(define (ith-derivative-by-tower i f)
 (lambda (c) (let ((e (generate-epsilon))) (univariate-e e i (f (linear-term e c 1))))))

(define (position-of-nonzero i)
 (cond ((null? i) #f)
       ((zero? (car i)) (let ((position (position-of-nonzero (cdr i)))) (if position (+ position 1) #f)))
       (else 0)))

(define (decrement-lth i l) (if (zero? l) (cons (- (car i) 1) (cdr i)) (cons (car i) (decrement-lth (cdr i) (- l 1)))))

(define (list-replace-lth c l u) (if (zero? l) (cons u (cdr c)) (cons (car c) (list-replace-lth (cdr c) (- l 1) u))))

(define (partial-derivative-by-repetition i f)
 (let ((l (position-of-nonzero i)))
  (cond ;; Equation (8)
        ((not l) f)
        ;; Equation (9)
        (else (partial-derivative-by-repetition
               (decrement-lth i l) (lambda (c) ((derivative (lambda (u) (f (list-replace-lth c l u)))) (list-ref c l))))))))

;;; Equation (12)
(define (partial-derivative-by-tower i f)
 (lambda (c)
  (let ((e (map (lambda (cl) (generate-epsilon)) c)))
   (multivariate-e e i (f (map (lambda (el cl) (linear-term el cl 1)) e c))))))

Figure 4. A SCHEME implementation of D, the repetition and tower methods for D^i, and the repetition and tower methods for D^{(i1,...,in)}.

7. Discussion

Forward AD is typically formulated using very different machinery than that used above. The univariate first-order case is usually formulated as a transformation of a program whose program points compute values f(x) of the program input x to one whose program points compute values ⟨f(x), f^′(x)⟩. Since the program is a composition of primitives, the program transformation, as well as the transformation of the primitives, are formulated as applications of the chain rule. Karczmarczuk formulated the univariate higherorder case as a similar transformation to a program whose program points compute stream values (f(x), f^′(x), f^{′′}(x), . . .) via the chain rule, though he did not present a derivation of the transformations of the primitives.

The streams we have used are similar, but contain factorial factors: (f(x)/0!, f^′(x)/1!, f^{′′}(x)/2!, . . . , f^{(i)}(x)/i!, . . .). These Taylor series streams simplify some bookkeeping, in particular allowing the use of Taylor expansions instead of the chain rule in the derivations. This makes the multivariate higher-order case more straightforward to derive and justify. However, since each representation can be converted to the other (using operators that are similar to C), we do not consider this a fundamental difference.

Karczmarczuk [4] hinted at a formulation of the multivariate higher-order case using the chain rule, where lazy unidimensional streams are replaced with lazy trees, but did not present a derivation or justification of the method’s correctness. That method redundantly represents identical cross derivatives, i.e., ∂^2f/∂xi∂xj = ∂^2f/∂xj ∂xi. Our method avoids that inefficiency. Moreover, although nesting is not our topic, the code presented does allow the derivative-taking constructs to nest correctly.

Laziness is particularly useful when representing and manipulating power series, in contexts beyond those considered here [10]. For instance it can be used to define a power series with a recurrence relation. Such power series arise naturally in related contexts, such as differential equations that cannot be solved in closed form. Formulating nestable multivariate higher-order forward AD in terms of lazy power-series representations can allow forward AD to inter-operate with such other applications of power series.

Acknowledgments

This work was supported, in part, by NSF grant CCF-0438806, Science Foundation Ireland grant 00/PI.1/C067, and a grant from the Higher Education Authority of Ireland. Any opinions, findings, and conclusions or recommendations expressed in this material are those of the author(s) and do not necessarily reflect the views of the National Science Foundation.

References

[1] R. E. Wengert, “A simple automatic derivative evaluation program,” Comm. of the ACM, vol. 7, no. 8, pp. 463–4, 1964.

[2] J. Karczmarczuk, “Functional differentiation of computer programs,” in Proceedings of the III ACM SIGPLAN International Conference on Functional Programming, Baltimore, MD, Sept. 1998, pp. 195–203.

[3] ——, “Lazy differential algebra and its applications,” in Workshop, III International Summer School on Advanced Functional Programming, Braga, Portugal, Sept. 1998.

[4] ——, “Functional coding of differential forms,” in Scottish Workshop on FP, Sept. 1999.

[5] ——, “Functional differentiation of computer programs,” Journal of Higher-Order and Symbolic Computation, vol. 14, pp. 35–57, 2001.

[6] J. M. Siskind and B. A. Pearlmutter, “Nesting forward-mode AD in a functional framework,” to appear.

[7] B. Taylor, Methodus Incrementorum Directa et Inversa. London, 1715.

[8] W. K. Clifford, “Preliminary sketch of bi-quaternions,” Proceedings of the London Mathematical Society, vol. 4, pp. 381– 395, 1873.

[9] W. G. Horner, “A new method of solving numerical equations of all orders, by continuous approximation,” Philos. Trans. Roy. Soc. London, vol. 109, pp. 308–335, July 1819.

[10] J. Karczmarczuk, “Generating power of lazy semantics,” Theoretical Computer Science, vol. 187, 1997.